

Using GPS to accurately establish True Airspeed (TAS).

June, 1998

Doug Gray
5 Larkspur Place, Heathcote, NSW Australia 2233
Phone/Fax 61 2 9520 5542
Email: dgra1233@bigpond.net.au

Background

True Airspeed or TAS and its determination is an essential part in the assessment of aircraft performance. From airspeed indicator calibration to performance evaluation it is necessary to know or determine TAS accurately.


The following describes yet another method for determining TAS but in this case it allows a little more flexibility in flying the aircraft. Cramped airspaces should no longer be a limiting factor.

Using the GPS

One very good technique¹ described by David Fox requires the aircraft to be flown accurately over specific ground tracks, with the recorded speeds combined in a simple computation to derive the TAS. In fact this is a special case of the general solution to the problem.

The general solution still requires three legs but these can be (almost) any heading/track. The aviating will be more convenient, but the calculation will be more difficult, hence I recommend a spreadsheet. After noting down the GPS Speed plus the GPS Track for each leg, these six figures are entered to arrive at the TAS.

The following illustration is helpful to understand the geometry of what we are doing. You will see the drawing shows the three vector triangles corresponding to the three legs each superimposed onto the same wind speed vector.



Vector Triangles for three G/S tracks.

¹KITPLANES Feb 1995 page 49

G/S_1 , G/S_2 and G/S_3 correspond to the recorded ground speed vectors. If the Wind speed and direction are constant for the legs then the velocity vector triangles corresponding to the recorded ground speeds can be drawn on top of this single common Wind Speed vector W/S . The tails of the G/S vectors each will lie on a circle with a radius corresponding to the True Airspeed.

Since only three points are required to define the circle, three tracks are required to give a solution for TAS.

Computation

Rather than providing a derivation I have reduced the solution to a series of spreadsheet equations so that it may be of some practical value. These Excel equations can probably be extended to any alternate spreadsheet software without too much difficulty.

The wind direction result is the direction from which the wind is coming just like the weather report. The vector direction of the wind is at 180° from this.

	A	B	C
1	Speed1	140	140
2	Track1	192	192
3	Speed2	112	112
4	Track2	283	283
5	Speed3	120	120
6	Track3	20	20
7	X1	=B1*SIN(PI()*(360-B2)/180)	
8	Y1	=B1*COS(PI()*(360-B2)/180)	
9	X2	=B3*SIN(PI()*(360-B4)/180)	
10	Y2	=B3*COS(PI()*(360-B4)/180)	
11	X3	=B5*SIN(PI()*(360-B6)/180)	
12	Y3	=B5*COS(PI()*(360-B6)/180)	
13	M1	=-1*(B9-B7)/(B10-B8)	
14	B1	=(B8+B10)/2-B13*(B7+B9)/2	
15	M2	=-1*(B11-B7)/(B12-B8)	
16	B2	=(B8+B12)/2-B15*(B7+B11)/2	
17	WX	=(B14-B16)/(B15-B13)	
18	WY	=B13*B17+B14	
19	Wind Speed	=SQRT(B17^2+B18^2)	20.6
20	Wind Direction	=MOD(540-(180/PI()*ATAN2(B18,B17)),360)	314.8
21	TAS	=SQRT((B7-B17)^2+(B8-B18)^2)	130
22	Heading 1	=MOD(540-(180/PI()*ATAN2(B18-B8,B17-B7)),360)	200
23	Heading 2	=MOD(540-(180/PI()*ATAN2(B18-B10,B17-B9)),360)	287.8
24	Heading 3	=MOD(540-(180/PI()*ATAN2(B18-B12,B17-B11)),360)	11.7

Accuracy

Needless to say the aircraft should be flown on a constant heading with the speed stabilised, and each recorded at the same power setting and same altitude. It would be good form to ensure the aircraft was not in a slight yaw by making sure the slip ball is centred.

The actual tracks used are not critical, in fact any three tracks will give a TAS figure, but if the headings are too close to each other then the TAS will be subject to error.

With a smaller difference in heading the greater will be the error in the final solution. This can be appreciated by looking at fitting the circle to three points. The closer these three points are then the more variability we will see in the size and position of the circle due to our measurement errors.

Roughly speaking provided the three tracks differ from somewhere about 90° to 120° then the resulting error in TAS will be of the same magnitude as the error in the GPS Ground

Speed. An error in GPS speed of ± 1 knot and $\pm 1^\circ$ in track will result in a corresponding error of about ± 1.3 knot.

Do not forget that other factors will also affect the accuracy. The most difficult to eliminate is as a result of being in an ascending or descending airmass. This effect could be minimised by testing in stable atmospheric conditions, or by collecting sufficient data points so as to average out this error.

Determine your Compass Deviation as well.

In the spreadsheet equations I have included three lines titled 'Heading'. These correspond to the actual headings for each of the legs or in other words the direction for each of the TAS vectors.

If the compass heading is noted with the GPS data then over a period of aircraft assessment plotting compass heading against the corresponding computed TAS Heading will enable a deviation chart to be compiled for the aircraft. This of course assumes that the GPS is set to display Magnetic track data.

While a conventional compass swing will provide such data very quickly and conveniently, the opportunity to get such data for nothing is hard to resist.